QR decomposition:

Every matrix A can be written as $A = Q.R$ where
(1) Q is an orthogonal matrix, namely $Q^T Q = I$
(2) R is an upper-triangular matrix

For example:

```
> A := LinearAlgebra:-RandomMatrix(4,3,generator=-5..5);
```

$$A := \begin{bmatrix} 1 & 0 & -2 \\ 0 & -5 & 4 \\ -2 & -4 & 3 \\ 1 & -5 & -2 \end{bmatrix} \quad (1)$$

```
> (Q,R) := LinearAlgebra:-QRDecomposition(A,fullspan);
```

$$Q, R := \begin{bmatrix} \frac{1}{6}\sqrt{6} & -\frac{1}{258}\sqrt{258} & -\frac{20}{65747}\sqrt{197241} & \frac{61}{4587}\sqrt{4587} \\ 0 & -\frac{5}{129}\sqrt{258} & \frac{346}{197241}\sqrt{197241} & \frac{5}{4587}\sqrt{4587} \\ -\frac{1}{3}\sqrt{6} & -\frac{1}{43}\sqrt{258} & -\frac{145}{197241}\sqrt{197241} & \frac{20}{4587}\sqrt{4587} \\ \frac{1}{6}\sqrt{6} & -\frac{11}{258}\sqrt{258} & -\frac{230}{197241}\sqrt{197241} & -\frac{7}{1529}\sqrt{4587} \end{bmatrix}, \begin{bmatrix} \sqrt{6} & \frac{1}{2}\sqrt{6} & -\frac{5}{3}\sqrt{6} \\ 0 & \frac{1}{2}\sqrt{258} & -\frac{17}{129}\sqrt{258} \\ 0 & 0 & \frac{1}{129}\sqrt{197241} \\ 0 & 0 & 0 \end{bmatrix} \quad (2)$$

```
> LinearAlgebra:-Transpose(Q).Q;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (3)$$

```
> Q.LinearAlgebra:-Transpose(Q);
```

(4)

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(4)}$$

```
> B := A:
  B[1..-1,2] := A[1..-1,3]:
  B;
```

$$\begin{bmatrix} 1 & -2 & -2 \\ 0 & 4 & 4 \\ -2 & 3 & 3 \\ 1 & -2 & -2 \end{bmatrix} \quad \textbf{(5)}$$

```
> (U,S) := LinearAlgebra:-QRDecomposition(B,fullspan);
```

$$U, S := \begin{bmatrix} \frac{1}{6}\sqrt{6} & -\frac{1}{21}\sqrt{3} & \frac{9}{14}\sqrt{2} & 0 \\ 0 & \frac{4}{7}\sqrt{3} & \frac{4}{63}\sqrt{2} & \frac{1}{9} \\ -\frac{1}{3}\sqrt{6} & -\frac{1}{21}\sqrt{3} & \frac{16}{63}\sqrt{2} & \frac{4}{9} \\ \frac{1}{6}\sqrt{6} & -\frac{1}{21}\sqrt{3} & -\frac{17}{126}\sqrt{2} & \frac{8}{9} \end{bmatrix}, \begin{bmatrix} \sqrt{6} & -\frac{5}{3}\sqrt{6} & -\frac{5}{3}\sqrt{6} \\ 0 & \frac{7}{3}\sqrt{3} & \frac{7}{3}\sqrt{3} \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(6)}$$

```
> C := LinearAlgebra:-RandomMatrix(2,4,generator=-5..5);
```

$$C := \begin{bmatrix} -5 & 1 & 0 & -2 \\ 1 & 5 & -4 & -1 \end{bmatrix} \quad \textbf{(7)}$$

```
> (W,P) := LinearAlgebra:-QRDecomposition(C,fullspan);
```

$$W, P := \begin{bmatrix} -\frac{5}{26}\sqrt{26} & \frac{1}{26}\sqrt{26} \\ \frac{1}{26}\sqrt{26} & \frac{5}{26}\sqrt{26} \end{bmatrix}, \begin{bmatrix} \sqrt{26} & 0 & -\frac{2}{13}\sqrt{26} & \frac{9}{26}\sqrt{26} \\ 0 & \sqrt{26} & -\frac{10}{13}\sqrt{26} & -\frac{7}{26}\sqrt{26} \end{bmatrix} \quad \textbf{(8)}$$

```
>
```

Example of the Householder transformation

```
> u, v := <1,2,3>, <sqrt(14),0,0>;
```

(9)

$$u, v := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}, \begin{bmatrix} \sqrt{14} \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(9)}$$

```
> y := u-v;
```

$$y := \begin{bmatrix} 1 - \sqrt{14} \\ 2 \\ 3 \end{bmatrix} \quad \textbf{(10)}$$

```
> w := simplify(y/LinearAlgebra:-Norm(y,2));
```

$$w := \begin{bmatrix} -\frac{-1 + \sqrt{14}}{\sqrt{28 - 2\sqrt{14}}} \\ \frac{2}{\sqrt{28 - 2\sqrt{14}}} \\ \frac{3}{\sqrt{28 - 2\sqrt{14}}} \end{bmatrix} \quad \textbf{(11)}$$

```
> H := LinearAlgebra:-IdentityMatrix(3)-2*w.LinearAlgebra:-
Transpose(w);
```

$$H := \begin{bmatrix} 1 - \frac{2\left(-1 + \sqrt{14}\right)^2}{28 - 2\sqrt{14}} & \frac{4\left(-1 + \sqrt{14}\right)}{28 - 2\sqrt{14}} & \frac{6\left(-1 + \sqrt{14}\right)}{28 - 2\sqrt{14}} \\ \frac{4\left(-1 + \sqrt{14}\right)}{28 - 2\sqrt{14}} & 1 - \frac{8}{28 - 2\sqrt{14}} & -\frac{12}{28 - 2\sqrt{14}} \\ \frac{6\left(-1 + \sqrt{14}\right)}{28 - 2\sqrt{14}} & -\frac{12}{28 - 2\sqrt{14}} & 1 - \frac{18}{28 - 2\sqrt{14}} \end{bmatrix} \quad \textbf{(12)}$$

```
> simplify(H.u);
```

$$\begin{bmatrix} \sqrt{14} \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(13)}$$

Demonstration of QR decomposition

```
> A := LinearAlgebra:-RandomMatrix(4,3,generator=-5..5);
```

$$A := \begin{bmatrix} 5 & -3 & 0 \\ -1 & 5 & 5 \\ -3 & 0 & -4 \\ 2 & 1 & -3 \end{bmatrix} \quad (14)$$

```
> s := LinearAlgebra:-Norm(A[1..-1,1],2);
```

$$s := \sqrt{39} \quad (15)$$

```
> y := A[1..-1,1]-<s,0,0,0>:
  w := y/LinearAlgebra:-Norm(y,2):
  H1 := LinearAlgebra:-IdentityMatrix(4)-2*w.LinearAlgebra:-
  Transpose(w):
> R := simplify(H1.A);
```

$$R := \begin{bmatrix} -\frac{39\left(-5+\sqrt{39}\right)}{-39+5\sqrt{39}} & \frac{18\left(-5+\sqrt{39}\right)}{-39+5\sqrt{39}} & -\frac{-5+\sqrt{39}}{-39+5\sqrt{39}} \\ 0 & \frac{-177+22\sqrt{39}}{-39+5\sqrt{39}} & \frac{-196+25\sqrt{39}}{-39+5\sqrt{39}} \\ 0 & -\frac{9\left(-6+\sqrt{39}\right)}{-39+5\sqrt{39}} & -\frac{-153+20\sqrt{39}}{-39+5\sqrt{39}} \\ 0 & \frac{-75+11\sqrt{39}}{-39+5\sqrt{39}} & -\frac{-119+15\sqrt{39}}{-39+5\sqrt{39}} \end{bmatrix} \quad (16)$$

```
> s := simplify(LinearAlgebra:-Norm(R[2..-1,2],2));
```

$$s := \frac{\sqrt{347}\left(-3\sqrt{13}+5\sqrt{3}\right)}{-39+5\sqrt{3}\sqrt{13}} \quad (17)$$

```
> y := R[2..-1,2]-<s,0,0>:
  w := y/LinearAlgebra:-Norm(y,2):
  T := LinearAlgebra:-IdentityMatrix(3)-2*w.LinearAlgebra:-
  Transpose(w):
  H2 := LinearAlgebra:-IdentityMatrix(4)[1..-1,1..-1]:
  H2[2..4,2..4] := T:
> R2 := H2.R:
> evalf(R2);
```

$$\begin{bmatrix} 6.244997994 & -2.882306767 & 0.1601281537 \\ 0. & 5.166459874 & 4.347568655 \\ 0. & -3.98\ 10^{-8} & -0.534577894 \\ 0. & -1.134\ 10^{-7} & 5.548624346 \end{bmatrix} \quad (18)$$

```
> s := LinearAlgebra:-Norm(R2[3..4,3],2):
```

```
> y := R2[3..4,3]-<s,0>:
  w := y/LinearAlgebra:-Norm(y,2):
  T:= LinearAlgebra:-IdentityMatrix(2)-2*w.LinearAlgebra:-
  Transpose(w):
  H3 := LinearAlgebra:-IdentityMatrix(4)[1..-1,1..-1]:
  H3[3..4,3..4] := T:
> R3 := H3.R2:
> evalf(R3);
```

$$\begin{bmatrix} 6.244997994 & -2.882306767 & 0.1601281537 \\ 0. & 5.166459874 & 4.347568655 \\ 0. & -1.090605079\ 10^{-7} & 5.574316607 \\ 0. & -5.049163904\ 10^{-8} & 1.1\ 10^{-9} \end{bmatrix} \quad (19)$$

```
> Q := H1.H2.H3:
> evalf(A-Q.R3);
```

$$\begin{bmatrix} 0. & -9.0715279\ 10^{-10} & 2.736076013\ 10^{-9} \\ -8.\ 10^{-10} & 5.67312799\ 10^{-9} & 9.434244211\ 10^{-9} \\ -2.\ 10^{-9} & -9.2770575\ 10^{-10} & -2.259950019\ 10^{-9} \\ 2.\ 10^{-9} & 7.1288732\ 10^{-10} & -2.262992957\ 10^{-9} \end{bmatrix} \quad (20)$$

```
>
```